

## Using SAS's PROC GPLOT to plot data and lines

PROC GPLOT creates “publication quality” color graphics which can easily be exported into documents, presentations, etc. To export the graphs for future use click on **file, export**. In the dialog box choose a file name and file type (\*.emf works well for putting graphs in word documents or powerpoint presentations)

- 1 **proc gplot options;**
- 2 **title n options 'text';**
  - 2.a **n** = title level 1-10
  - 2.b **c** = color (black, red, green, blue, yellow, cyan, etc)
  - 2.c **f** = font (swiss, simplex, script, italic, zapf, etc.)
  - 2.d **h** = height (decimal value, eg 1, .5, 2, etc.)
- 3 **symbol n options;**
  - 3.a **n** = symbol number 1 to 255
  - 3.b **c** = color (black, red, green, blue, yellow, cyan, etc) of symbol and interpolation line.
    - 3.b.1 **cv** = color of symbol
    - 3.b.2 **ci** = color of interpolation line
  - 3.c **f** = font (swiss, simplex, script, italic, zapf, etc.)
  - 3.d **w** = width (in pixels, e.g., 1, 2, 3, etc.)
  - 3.e **h** = height (in cells, decimal value, eg 1, .5, 2, etc.)
  - 3.f **l** = line (type line, 1=solid, 2-46 various dashed lines)
  - 3.g **v** = value (type of symbol, e.g. plus, x, star, square, diamond, triangle, dot, circle)
  - 3.h **i** = interpolation option
    - 3.h.1 **i=none** no line connecting points
    - 3.h.2 **i = join** points connected by a straight line
    - 3.h.3 **i = spline** – plot line interpolated using a cubic spine
- 4 **label y = “ylabel” x = “xlabel”;** renames variables as they appear on graph.
- 5 **plot yvariable\*xvariable=n .../ options;**
  - 5.a **plot y\*x=n;** plots y against x using symbol n
  - 5.b **plot y\*x=n1 r\*s=n2;** creates 2 (or more) separate plots.
  - 5.c **plot y\*x=n1 r\*s=n2/overlay;** puts plots on same graph.
  - 5.d **plot y\*x=n1 r\*s=n2/overlay legend;** puts plots on same graph and adds legend.
  - 5.e **plot y\*x=n/href=5;** plots y against x using symbol n and puts *vertical reference line at x=5*.
  - 5.f **plot y\*x=n/href=5 lh=linetype ch = color;** lh specifies type of href line (1-46), ch specifies color of href line. Can use **vref=c** to specify horizontal reference line, **vh** specifies type of vref line, **cv** specifies color of vref line.

*Some Gplot examples using the food expenditure data*

```
data food;
infile 'table5.2';
input y x;
proc reg;
model y = x/p;
output out=foodout p=yhat l95=lb u95=ub r=ehat;
```

*Read data, estimate regression, obtain predicted values, create output dataset containing predicted values, 95% confidence bands and residuals.*

\* Example 1;

```
symbol1 v=dot c=red i=none h=.5;
```

```
proc gplot data=foodout;
label y="food expenditure" x="income";
plot y*x=1;
title h=2 f=swiss "Example 1";
title2 h=1.5 f=italic c=red "Scatter diagram";
run;
```

*Example 1 plots data only using symbol1. label statement changes what appears on axes.*

\* Example 2;

```
symbol2 v=dot c=red i=none h=.5;
symbol3 c=black i=join l=1 w=4;
proc gplot;
plot y*x=2 yhat*x=3/overlay legend;
title h=2 f=swiss "Example 2";
title2 h=1.5 f=italic c=red "Scatter diagram and regression line";
```

*Example 2 plots data using symbol2 and fitted regression line using symbol3*

\* Example 3;

```
symbol4 v=dot c=red i=none h=.5;
symbol5 c=black i=join l=1;
symbol6 c=green i=join l=2;
proc gplot;
plot y*x=4 yhat*x=5 lb*x=6 ub*x=6/overlay legend;
title h=2 f=swiss "Example 3";
title2 h=1.5 f=italic c=red "Scatter diagram and regression line";
title3 h=1.5 f=italic c=red "Confidence bands included";
```

*Example 3 adds the confidence bands using symbol6*

\* Example 4;


```
symbol4 v=dot c=red i=none h=.5;
symbol5 c=black i=join l=1;
symbol6 c=green i=join l=2;
proc gplot;
plot y*x=4 yhat*x=5 lb*x=6 ub*x=6/overlay legend href=750 ch=purple lh=3;
title h=2 f=swiss "Example 4";
title2 h=1.5 f=italic c=red "Scatter diagram and regression line";
title3 h=1.5 f=italic c=red "Confidence bands & reference line included";

run;
```

*Example 4 adds vertical reference line at x=750, in the color purple, using linetype = 3*


### Example 1

*Scatter diagram*




### Example 2

*Scatter diagram and regression line*




### Example 3

*Scatter diagram and regression line  
Confidence bands included*



### Example 4

*Scatter diagram and regression line  
Confidence bands & reference line included*



01/21/04

The graphs above appear in color on your screen, and if you have a color printer they will come out in color when printed in a document, etc.

In addition to the options illustrated above there are options to add footnotes to the graph, notes to the graph, of course controlling color, size, position, etc.

SAS/GRAPH can produce

- bar charts
- pie charts
- maps
- 3-dimensional graphs

For many more additional examples use SAS's HELP menu.

- while in SAS click on HELP
- choose SAMPLE PROGRAMS
- double-click on SAS SAMPLE LIBRARY on the CONTENTS tab
- click on SAS/GRAPH
- select an example.

You can copy the sample code and paste it into the program window. Execute the file to see what the program can do. Many of these programs do quite tricky stuff. Don't get bogged down in the details.

You can also visit the SAS web site: try <http://www.sas.com/techsup/download/sample/README> for a list of samples, of various types, which you can download, not only for SAS/GRAPH but for regression analysis and other topics.